PHYS 301
HOMEWORK #5
Solutions

On this homework assignment, you may use Mathematica to compute integrals, but you must submit your Mathematica output with your assignment.

1. For $f(x) = \begin{cases} -1, & -1 < x < 0 \\ 1, & 0 < x < 1 \end{cases}$

Find the Fourier coefficients and write out the first three non - zero terms of the series expansion.

Solution : This is an odd function on (-1, 1), so we know that the a coefficients are zero. Since the function is 2 L = 2 periodic, L = 1. Computing the b coefficients :

$$b_n = \frac{1}{1}\int_{-1}^{1} f(x)\sin(n\pi x)\,dx = 2\int_0^1 1\cdot\sin(n\pi x)\,dx = \frac{-2}{n\pi}\cos(n\pi x)\Big|_0^1 =$$

$$\frac{-2}{n\pi}(\cos(n\pi)-1) = \frac{2}{n\pi}(1-(-1)^n) = \begin{cases} \frac{4}{n\pi}, & \text{odd} \\ 0, & \text{even} \end{cases}$$

Therefore, our Fourier series is :

$$f(x) = \frac{4}{\pi}\left[\sin\pi x + \frac{\sin 3\pi x}{3} + \sin\frac{5\pi x}{5} + ...\right]$$

Verifying via Mathematica :

```
Plot[(4 / π) Sum[Sin[n π x] / n, {n, 1, 31, 2}], {x, -3, 3}]
```



---

2. For $f(x) = \begin{cases} x, & 0 < x < 1 \\ 2 - x^2, & 1 < x < 2 \end{cases}$

extend f to construct a) an odd function on (-2, 2) and b) an even function on (-2, 2). Compute the Fourier coefficients for each series and write out the first three non - zero terms of each expansion. (20 pts for this problem).

Solution : a) If we construct an odd function on (-2, 2), we know that only the b_n coefficients will be non - zero; we compute these via :

$$b_n = \frac{2}{2}\int_0^2 f(x)\,dx = \int_0^1 x\sin(n\pi x/2)\,dx + \int_1^2 \left(2-x^2\right)\sin(n\pi x/2)\,dx$$

We obtain :

```
In[2]:= Integrate[x Sin[n π x / 2], {x, 0, 1}] + Integrate[(2 - x^2) Sin[n π x / 2], {x, 1, 2}]
```

$$\text{Out[2]= } \frac{-2 n\pi \operatorname{Cos}\left[\frac{n\pi}{2}\right] + 4\operatorname{Sin}\left[\frac{n\pi}{2}\right]}{n^2\pi^2} + \frac{2\left(\left(8+n^2\pi^2\right)\operatorname{Cos}\left[\frac{n\pi}{2}\right] + 2\left(-4+n^2\pi^2\right)\operatorname{Cos}[n\pi] + 4 n\pi\left(\operatorname{Sin}\left[\frac{n\pi}{2}\right] - 2\operatorname{Sin}[n\pi]\right)\right)}{n^3\pi^3}$$

Simplifying this output using the fact that n is an integer :

```
In[3]:= Simplify[%, Assumptions → n ∈ Integers]
```

$$\text{Out[3]= } \frac{4\left((-1)^n\left(-4+n^2\pi^2\right) + 4\operatorname{Cos}\left[\frac{n\pi}{2}\right] + 3 n\pi\operatorname{Sin}\left[\frac{n\pi}{2}\right]\right)}{n^3\pi^3}$$

If we examine this expression, we see that the argument of the trig functions (n π/2) indicates we will need to consider 4 separate cases (i.e., 4 separate values) of n.

Since we have terms involving sin and cos of n π/2, we know that we need to consider cases where n = {1, 2, 3, 4}. We can use the Mod command successively in conjuction with Assumptions :

```
In[4]:= Simplify[%, Assumptions → Mod[n, 4] == 1]
```

$$\text{Out[4]= } \frac{4\left(4+3 n\pi - n^2\pi^2\right)}{n^3\pi^3}$$

(These are the coefficients when n = 1, 5, 9, ...)

```
In[5]:= Simplify[%%, Assumptions → Mod[n, 4] == 2]
```

$$\text{Out[5]= } \frac{4\left(-8+n^2\pi^2\right)}{n^3\pi^3}$$

(These are the coefficients when n = 2, 6, 10, ...)

```
In[6]:= Simplify[%%%, Assumptions → Mod[n, 4] == 3]
```

$$\text{Out[6]= } -\frac{4\left(-4+3 n\pi + n^2\pi^2\right)}{n^3\pi^3}$$

(These are the coefficients for n = 3, 7, 11, ...)

In[7]:=
```
Simplify[%%%%, Assumptions → Mod[n, 4] == 0]
```

Out[7]= $\frac{4}{n\pi}$

(Finally, these are the coefficients for n = 4, 8, 12, ...)

Using these data, the first few coefficients are :

$$b_1 = \frac{16 + 3\pi - \pi^2}{\pi^3}$$

$$b_2 = \frac{-32 + 16\pi^2}{8\pi^3}$$

$$b_3 = \frac{16 - 36\pi - 36\pi^2}{27\pi^3}$$

$$b_4 = \frac{1}{\pi}$$

And we can write the first few non - zero terms of the expansion, using these coefficients, as :

$$f(x) = b_1 \sin(\pi x/2) + b_2 \sin(2\pi x/2) + b_3 \sin(3\pi x/2) + b_4 \sin(4\pi x/2)$$

You can verify these results by writing the following short code :

In[67]:=
```
Clear[bn, f]
bn = Integrate[x Sin[n π x / 2], {x, 0, 1}] + Integrate[(2 - x^2) Sin[n π x / 2], {x, 1, 2}];
f = Which[0 < x < 1, x, 1 < x < 2, 2 - x^2];
Plot[{f, Sum[bn Sin[n π x / 2], {n, 21}]}, {x, -2, 2}]
```

Out[70]=



b) For the even solution, we extend f as an even function on (-2, 2). We can employ symmetry to argue that the b_n coefficients are zero, and that the a_n coefficients can be determined from :

$$a_0 = \frac{2}{2}\int_0^2 f(x)\,dx = \int_0^1 x\,dx + \int_1^2 \left(2 - x^2\right) dx = \frac{1}{6}$$

We use Mathematica to compute the a_n coefficients and plot the function :

In[144]:=
```
Clear[a0, an, f]
f = Which[0 < x < 1, x, 1 < x < 2, 2 - x^2];
a0 = Integrate[x, {x, 0, 1}] + Integrate[2 - x^2, {x, 1, 2}];
an = Integrate[f Cos[n π x / 2], {x, 0, 2}];
Print["a0 = ", a0, ";  ", "an = ", an]
Print["The first five terms of the Fourier expansion are: ",
 a0 / 2 + Sum[an Cos[n π x / 2], {n, 5}]]
Plot[a0 / 2 + Sum[an Cos[n π x / 2], {n, 21}], {x, -2, 2}]
```

$$a0 = \frac{1}{6}; \quad an = -\frac{4\left(n\pi - 3n\pi \operatorname{Cos}\left[\frac{n\pi}{2}\right] + 4n\pi \operatorname{Cos}[n\pi] + 4\operatorname{Sin}\left[\frac{n\pi}{2}\right] - 4\operatorname{Sin}[n\pi] + n^2\pi^2\operatorname{Sin}[n\pi]\right)}{n^3\pi^3}$$

The first five terms of the Fourier expansion are:

$$\frac{1}{12} - \frac{4(4-3\pi)\operatorname{Cos}\left[\frac{\pi x}{2}\right]}{\pi^3} - \frac{8\operatorname{Cos}[\pi x]}{\pi^2} - \frac{4(-4-9\pi)\operatorname{Cos}\left[\frac{3\pi x}{2}\right]}{27\pi^3} - \frac{\operatorname{Cos}[2\pi x]}{2\pi^2} - \frac{4(4-15\pi)\operatorname{Cos}\left[\frac{5\pi x}{2}\right]}{125\pi^3}$$

And the plot of this Fourier expansion :

Out[150]=



---

3. Problem 24, p. 371 of the text.

Solution : The first thing we need to do to solve this problem is write the equation of the string on (0, L). Using simple geometry, we have :

$$f(x) = \begin{cases} 4hx/L, & 0 < x < L/4 \\ 2h - 4hx/L, & L/4 < x < L/2 \\ 0, & L/2 < x < L \end{cases}$$

Knowing we are constructing a sin series, we know we need only worry about the b coefficients, and we can calculate them via :

$$b_n = \frac{1}{L}\int_{-L}^{L} f(x)\sin(n\pi x/L)\,dx = \frac{2}{L}\int_{0}^{L} f(x)\sin(n\pi x/L)\,dx =$$

In[161]:=
```
2/L (Integrate[(4 h x / L) Sin[n π x / L], {x, 0, L / 4}] +
   Integrate[(2 h - 4 h x / L) Sin[n π x / L], {x, L / 4, L / 2}])
```

and when we do this integration we get :

$$\frac{2\left(\frac{h\,L\left(-n\,\pi\,\mathrm{Cos}\left[\frac{n\pi}{4}\right]+4\,\mathrm{Sin}\left[\frac{n\pi}{4}\right]\right)}{n^2\,\pi^2}+\frac{h\,L\left(n\,\pi\,\mathrm{Cos}\left[\frac{n\pi}{4}\right]+4\,\mathrm{Sin}\left[\frac{n\pi}{4}\right]-4\,\mathrm{Sin}\left[\frac{n\pi}{2}\right]\right)}{n^2\,\pi^2}\right)}{L}$$

Which by inspection becomes (well, with a little help from) :

In[162]:=
```
Simplify[%, Assumptions → n ∈ Integers]
```

Out[162]=
$$\frac{64\,h\,\mathrm{Cos}\left[\frac{n\,\pi}{8}\right]\mathrm{Sin}\left[\frac{n\,\pi}{8}\right]^3}{n^2\,\pi^2}$$

The nth term of the Fourier expansion is the nth b coefficient times Sin[n π x/L]. We can verify that these coefficients yield the proper curve by setting L and h to arbitrary but reasonable values (Mathematica won' t print symbolic functions) :

In[167]:=
```
Clear[f, bn, x, h, L]
h = 0.1; L = 1.0;
f = Which[0 < x < L / 4, 4 h x / L, L / 4 < x < L / 2, 2 h - 4 h x / L, L / 2 < x < L, 0];
bn = 2 / L (Integrate[4 h x Sin[ n π x / L] / L, {x, 0, L / 4}] +
     Integrate[(2 h - 4 h x / L) Sin[n π x / L], {x, L / 4, L / 2}]);
Plot[Sum[bn Sin[n π x / L], {n, 21}], {x, 0, L}]
```



---

4. Consider the following graph of one complete cycle of voltage vs. time :



corresponding to $V(t) = \begin{cases} 50, & 0 < t < 1/100 \\ -50, & 1/100 < t < 1/50 \end{cases}$

where V is measured in volts and t in seconds.

Write the Fourier series representing this pattern (assume this part of the graph is repeated 50 times/second).

Solution : Since the entire cycle lasts a duration of 1/50 s, the cycle is 2 L = 1/50 periodic, implying that L = 1/100 in this calculation. We use the familiar equations to find coefficients :

$$a_0 = \frac{1}{1/100}\int_0^{1/100} 50\,dt - \frac{1}{1/100}\int_{1/100}^{1/50} 50\,dt = 0$$

$$a_n = \frac{1}{1/100}\left[\int_0^{1/100} 50\cos[100\,n\,\pi\,t]\,dt - \int_{1/100}^{1/50} 50\cos[100\,n\,\pi\,t] = \right.$$

$$100\left[\frac{50}{100\,n\,\pi}\,\{\sin(100\,n\,\pi\,t)\Big|_0^{1/100} - \sin(100\,n\,\pi\,r)\Big|_{1/100}^{1/50}\right] =$$

$$\frac{50}{n\,\pi}\,(\sin(n\,\pi) - (\sin(2\,n\,\pi) - \sin(n\,\pi)) = 0$$

$$b_n = \frac{1}{1/100}\left[\int_0^{1/100} 50\sin[100\,n\,\pi\,t]\,dt - \int_{1/100}^{1/50} 50\sin[100\,n\,\pi\,t]\,dt = \right.$$

$$\frac{50}{n\,\pi}\left[-\cos(100\,n\,\pi\,t)\Big|_0^{1/100} + \cos(100\,n\,\pi\,t)\Big|_{1/100}^{1/50}\right. =$$

$$\frac{50}{n\,\pi}\left[-\{\cos(n\,\pi) - 1\} + (\cos(2\,n\,\pi) - \cos(n\,\pi)\}\right] = \frac{50}{n\,\pi}\left[2\,(1 - (-1)^n)\right] = \begin{cases} \frac{200}{n\,\pi}, & n\text{ odd} \\ 0, & n\text{ even} \end{cases}$$

$$V(t) = \frac{200}{\pi}\sum_{\text{odd}}^{\infty}\frac{\sin(100\,n\,\pi\,t)}{n} = \frac{200}{\pi}\left[\sin(100\,\pi\,t) + \frac{\sin(3\cdot 100\,\pi\,t)}{3} + \frac{\sin(5\cdot 100\,\pi\,t)}{5} + \ldots\right]$$

Verifying :

```
Plot[(200 / π) Sum[Sin[100 n π t] / n, {n, 1, 31, 2}], {t, 0, 3 / 50}]
```



5. Use the function in problem 1 of this assignment to find the value of

$$\sum_{\text{odd } n}^{\infty} \frac{1}{n^2}$$

Solution : We will use Parseval' s theorem :

$$\text{average value of } (f(x))^2 \text{ on } (-1, 1) = \left(\frac{a_0}{2}\right)^2 + \frac{1}{2}\sum_{n=1}^{\infty} a_n^2 + \frac{1}{2}\sum_{n-1}^{\infty} b_n^2$$

Since the length of this interval is 2, we write the average of f^2 as:

$$\frac{1}{2}\int_{-1}^{1} 1\, dx = 1$$

The Fourier series for this problem shows that all the a_n coefficients are zero, and that

$$b_n = \frac{4}{n\pi} \text{ for odd values of } n$$

Therefore,

$$\frac{1}{2}\sum_{\text{odd } n}^{\infty}\left(\frac{4}{n\pi}\right)^2 = \frac{8}{\pi^2}\sum_{\text{odd } n}^{\infty}\frac{1}{n^2} = 1 \Rightarrow \sum_{\text{odd } n}^{\infty}\frac{1}{n^2} = \frac{\pi^2}{8}$$

We could have figured this out from the result derived in class for the infinite sum of all the reciprocal squares, namely

$$\sum_{n=1}^{\infty}\frac{1}{n^2} = \frac{\pi^2}{6}$$

We can find the inifinite sum for the even numbers by noting that all even numbers are divisible by 2, so that we can represent them as 2 n. Thus, the sum of all even terms is simply :

$$\sum_{\text{even}}^{\infty}\frac{1}{n^2} = \sum_{n=1}^{\infty}\frac{1}{(2n)^2} = \frac{1}{4}\sum_{n=1}^{\infty}\frac{1}{n^2} = \frac{1}{4}\cdot\frac{\pi^2}{6} = \frac{\pi^2}{24}$$

The sum of the odd terms is just the difference between the sum over all numbers and the sum over the evens, or :

$$\sum_{n=1}^{\infty}\frac{1}{n^2}-\sum_{\text{even}}^{\infty}\frac{1}{n^2}=\sum_{\text{odd}}^{\infty}\frac{1}{n^2}=\frac{\pi^2}{6}-\frac{\pi^2}{24}=\frac{\pi^2}{8}\text{ as we found before.}$$

Finally, direct verification with Mathematica :

```
Sum[1 / n^2, {n, 1, ∞, 2}]
```

$$\frac{\pi^2}{8}$$

---

6. You are familiar with Fibonacci numbers (fib[n]). Let' s define a similar set of numbers, the so called Loyola U numbers which have the properties :

lu[0] = 2
lu[1] = 1
lu[n] = lu[n - 1] + lu[n - 2]

(Note that the lu numbers start at n = 0). Write a short Mathematica program to test the conjecture :

lu[n] = fib[n - 1] + fib[n + 1]

for n ≤ 30. If the conjecture is true for a given value of n, your program should output the ratio of lu[n]/fib[n]. If the conjecture is false for a given value of n, your program should print "The conjecture is false." Your printout should include both your program and all results. (20 pts for this question)

Solution :

We will define two functions, lu[n] and f[n] to describe the Loyola and Fibonacci numbers respectively. We will initialize values and describe functions :

```
Clear[lu, f]
f[1] = 1; f[2] = 1; lu[0] = 2; lu[1] = 1;
f[n_] := f[n] = f[n - 1] + f[n - 2]
lu[n_] := lu[n] = lu[n - 1] + lu[n - 2]
(*These definitions allow us to store previously
 computed values of lu[n] and f[n] to minimize computing time. *)
Do[If[lu[n] == f[n - 1] + f[n + 1], Print["For n = ", n, " The value of lu(n)/f(n) =  ",
   lu[n] / f[n] // N], Print["The conjecture is false."]], {n, 2, 30}]
(*We nest an If statement inside a Do loop; the If statement tests to see
 if the conjecture is true (make sure you notice that this requires
   the use of a double equal sign.  Note also the limits of the Do loop;
  what would happen (and why would it happen) if you started with n=1?)*)
(* Now execute the program: *)
```

```
For n = 2 The value of lu(n)/f(n) =  3.
For n = 3 The value of lu(n)/f(n) =  2.
For n = 4 The value of lu(n)/f(n) =  2.33333
For n = 5 The value of lu(n)/f(n) =  2.2
For n = 6 The value of lu(n)/f(n) =  2.25
For n = 7 The value of lu(n)/f(n) =  2.23077
For n = 8 The value of lu(n)/f(n) =  2.2381
For n = 9 The value of lu(n)/f(n) =  2.23529
For n = 10 The value of lu(n)/f(n) =  2.23636
For n = 11 The value of lu(n)/f(n) =  2.23596
For n = 12 The value of lu(n)/f(n) =  2.23611
For n = 13 The value of lu(n)/f(n) =  2.23605
For n = 14 The value of lu(n)/f(n) =  2.23607
For n = 15 The value of lu(n)/f(n) =  2.23607
For n = 16 The value of lu(n)/f(n) =  2.23607
For n = 17 The value of lu(n)/f(n) =  2.23607
For n = 18 The value of lu(n)/f(n) =  2.23607
For n = 19 The value of lu(n)/f(n) =  2.23607
For n = 20 The value of lu(n)/f(n) =  2.23607
For n = 21 The value of lu(n)/f(n) =  2.23607
For n = 22 The value of lu(n)/f(n) =  2.23607
For n = 23 The value of lu(n)/f(n) =  2.23607
For n = 24 The value of lu(n)/f(n) =  2.23607
For n = 25 The value of lu(n)/f(n) =  2.23607
For n = 26 The value of lu(n)/f(n) =  2.23607
For n = 27 The value of lu(n)/f(n) =  2.23607
For n = 28 The value of lu(n)/f(n) =  2.23607
For n = 29 The value of lu(n)/f(n) =  2.23607
For n = 30 The value of lu(n)/f(n) =  2.23607
```

Pretty clearly the conjecture seems reasonable. In fact lu numbers really exist and are known as Lucas Numbers.